
FORMATTED LISTING
; ####################################################################################################
; #   PROJECT  :  Apple /// SOS Parallel Printer Driver 1.31 (6502 Assembly Source Code)
; #   FILE NAME:  PARPRINT.text
; ####################################################################################################

000001                  .NOPATCHLIST
000002                  .TITLE       "SOS Parallel Printer Driver -- 1.31  17-Mar-83"
000003  ;-----------------------------------------------------------------------
000004  ;
000005  ;
000006  ;               SOS Parallel Printer Driver
000007  ;
000008  ;               Copyright (C) 1983 by Apple Computer Inc.
000009  ;               All rights reserved.
000010  ;
000011  ;
000012  ;       Revisions:
000013  ;
000014  ;       1.30    14-Jan-83
000015  ;               New driver design, replaces version 1.01 22-Sep-81.
000016  ;               Default slot number now undefined (was 1).
000017  ;               Defined DCB INVERT bit for printers with inverted logic.
000018  ;
000019  ;       1.31    17-Mar-83
000020  ;               Fixed test for ASC_CR in Auto Line Feed processing.
000021  ;
000022  ;-----------------------------------------------------------------------
000023
000024  DEVTYPE         .EQU         41                        ;Character device, output only, printer
000025  SUBTYPE         .EQU         03
000026  APPLE           .EQU         0001
000027  RELEASE         .EQU         1310
000028                  .PAGE
000029  ;-----------------------------------------------------------------------
000030  ;
000031  ;  The macro SWITCH performs an N way branch based on a switch index.  The
000032  ;  maximum value of the switch index is 127 with bounds checking provided
000033  ;  as an option.  The macro uses the A and Y registers and alters the C,
000034  ;  Z, and N flags of the status register, but the X register is unchanged.
000035  ;
000036  ;               SWITCH  [index], [bounds], adrs_table, [*]
000037  ;
000038  ;       index   This is the variable that is to be used as the switch index.
000039  ;               If omitted, the value in the accumulator is used.
000040  ;
000041  ;      bounds   This is the maximum allowable value for index.  If index
000042  ;               exceeds this value, the carry bit will be set and execution
000043  ;               will continue following the macro.  If bounds is omitted,
000044  ;               no bounds checking will be performed.
000045  ;
000046  ;  adrs_table   This is a table of addresses (low byte first) used by the
000047  ;               switch.  The first entry corresponds to index zero.
000048  ;
000049  ;           *   If an asterisk is supplied as the fourth parameter, the
000050  ;               macro will push the switch address but will not exit to
000051  ;               it; execution will continue following the macro.  The
000052  ;               program may then load registers or set the status before
000053  ;               exiting to the switch address.
000054  ;
000055  ;-----------------------------------------------------------------------
000056
000057                  .MACRO       SWITCH
000058                  .IF          "%1" <> ""                ;If PARM1 is present,
000059                  LDA          %1                        ;  Load A with switch index
000060                  .ENDC
000061                  .IF          "%2" <> ""                ;If PARM2 is present,
000062                  CMP          #%2+1                     ;  Perform bounds checking
000063                  BCS          $3579                     ;  on switch index
000064                  .ENDC
000065                  ASL          A
000066                  TAY
000067                  LDA          %3+1,Y                    ;Get switch address from table
000068                  PHA                                    ;  and push onto stack
000069                  LDA          %3,Y
000070                  PHA
000071                  .IF          "%4" <> "*"               ;If PARM4 is omitted,
000072                  RTS                                    ;  Exit to code
000073                  .ENDC                                  ;Otherwise, drop through
000074                  .IF          "%2" <> ""
000075  $3579
000076                  .ENDC
000077                  .ENDM
000078                  .PROC        PRINTER
000079                  .WORD        0FFFF
000080                  .WORD        68.
000081                  .ASCII       "Parallel Printer Driver -- "
000082                  .ASCII       "Copyright (C) 1983 by Apple Computer Inc."
000083  ;-----------------------------------------------------------------------
000084  ;

000085  ;       Device Handler Identification Block
000086  ;
000087  ;-----------------------------------------------------------------------
000088  IDBLK           .WORD        0000                      ;Link to next device handler
000089                  .WORD        LP_MAIN                   ;Entry point address
000090                  .BYTE        8                         ;Length of device name
000091                  .ASCII       ".PRINTER       "
000092                  .BYTE        80                        ;Device # (Mark active)
000093  SLOT            .BYTE        0FF                       ;Slot # of Parallel card
000094                  .BYTE        00                        ;Unit #
000095                  .BYTE        DEVTYPE
000096                  .BYTE        SUBTYPE
000097                  .BYTE        00
000098                  .WORD        0000
000099                  .WORD        APPLE
000100                  .WORD        RELEASE
000101  ;-----------------------------------------------------------------------
000102  ;
000103  ;       Device Handler Configuration Block
000104  ;
000105  ;-----------------------------------------------------------------------
000106                  .WORD        5
000107  RDYMASK         .BYTE        060                       ;Mask for valid ready bits
000108  RDYSTAT         .BYTE        040                       ;Printer ready status value
000109  ;--------+--------+--------+--------+--------+--------+--------+--------+
000110  ;  Power | Select |Pap. Out|Rib. Out|  Check |        |        |        |
000111  ;--------+--------+--------+--------+--------+--------+--------+--------+
000112  PROPTNS         .BYTE        00                        ;Printer options
000113  ;--------+--------+--------+--------+--------+--------+--------+--------+
000114  ; Invert | Auto LF|        |        |        |        |        |        |
000115  ;--------+--------+--------+--------+--------+--------+--------+--------+
000116  CTRLVAL         .BYTE        00                        ;Control word for UPIC
000117  ;--------+--------+--------+--------+--------+--------+--------+--------+
000118  ;    0   |    0   | STB pos|    0   | ACK pos| STB +8 | STB +4 | STB -2 |
000119  ;--------+--------+--------+--------+--------+--------+--------+--------+
000120  TIMEOUT         .BYTE        64                        ;Timeout count for ACK
000121                  .PAGE
000122  ;-----------------------------------------------------------------------
000123  ;
000124  ;       SOS Device Handler Interface
000125  ;
000126  ;-----------------------------------------------------------------------
000127
000128  SOSINT          .EQU         0C0
000129  REQCODE         .EQU         SOSINT+0                  ;SOS request code
000130  BUFFER          .EQU         SOSINT+2                  ;Buffer pointer
000131  REQCNT          .EQU         SOSINT+4                  ;Requested count
000132  CTLSTAT         .EQU         SOSINT+2                  ;Control/status code
000133  CSLIST          .EQU         SOSINT+3                  ;Control/status list pointer
000134
000135
000136  ;-----------------------------------------------------------------------
000137  ;
000138  ;       SOS Global Subroutines
000139  ;
000140  ;-----------------------------------------------------------------------
000141
000142  ALLOCSIR        .EQU         1913                      ;SOS resource allocation
000143  DEALCSIR        .EQU         1916                      ;SOS resource deallocation
000144  SYSERR          .EQU         1928                      ;SOS error return
000145
000146
000147  ;-----------------------------------------------------------------------
000148  ;
000149  ;       SOS Error Codes
000150  ;
000151  ;-----------------------------------------------------------------------
000152
000153  XREQCODE        .EQU         20                        ;Invalid request code
000154  XCTLCODE        .EQU         21                        ;Invalid control/status code
000155  XNOTOPEN        .EQU         23                        ;Device not open
000156  XNOTAVIL        .EQU         24                        ;Device not available
000157  XNORESRC        .EQU         25                        ;Resouce not available
000158  XBADOP          .EQU         26                        ;Invalid operation for device
000159
000160
000161  ;-----------------------------------------------------------------------
000162  ;
000163  ;       Misecllaneous Equates
000164  ;
000165  ;-----------------------------------------------------------------------
000166
000167  TRUE            .EQU         80
000168  FALSE           .EQU         00
000169  ASC_LF          .EQU         0A
000170  ASC_CR          .EQU         0D
000171  BITON7          .EQU         80
000172  BITOFF7         .EQU         7F
000173                  .PAGE
000174  ;-----------------------------------------------------------------------
000175  ;
000176  ;       Local Variables
000177  ;

000178  ;-----------------------------------------------------------------------
000179
000180  OPENFLG         .BYTE        FALSE                     ;Device open flag
000181  CRFLAG          .BYTE        FALSE                     ;Previous character was CR
000182  INVERT          .BYTE        0                         ;XOR with data for inverted logic
000183  DEVOFF          .BYTE        0                         ;Offset to I/O addresses
000184  CTRLWRD         .BYTE        0                         ;Parallel card control word
000185
000186  ;  Consumer State -- The Consumer State is the current operational status
000187  ;  of the interrupt handler.  The Idle state indicates that the consumer
000188  ;  has no data to send, and there is no pending ACK from the printer.  The
000189  ;  Waiting for ACK state indicates that data has been STROBEd but the ACK
000190  ;  is still pending.  The Waiting for Ready state indicates that data was
000191  ;  STROBEd when the printer was not Ready to accept it.  The printer must
000192  ;  ACK when it becomes Ready, then the consumer will STROBE again.
000193
000194  CSTATE          .BYTE        0                         ;Consumer State
000195  IDLE            .EQU         000                       ;  Idle, waiting for data
000196  WT_ACK          .EQU         080                       ;  Waiting for ACK
000197  WT_RDY          .EQU         0C0                       ;  Waiting for Ready
000198
000199
000200  ;  Local Buffer -- Logically, the Local Buffer contains $FF bytes and the
000201  ;  Buffer Count may range from $00 through $FF.  Physically, however, the
000202  ;  Local Buffer contains $100 bytes so the producer and consumer pointers
000203  ;  can be incremented without testing them for end of buffer or wrap around.
000204
000205  BUFCNT          .BYTE        0                         ;Local buffer byte count
000206  PRODPTR         .BYTE        0                         ;Producer buffer pointer
000207  CSMRPTR         .BYTE        0                         ;Consumer buffer pointer
000208  LOCBUF          .EQU         *                         ;Local buffer
000209                  .ASCII       "Copyright (C) 1983 by Apple Computer Inc."
000210                  .BLOCK       *-LOCBUF^0FF+1            ;Save $100 bytes for Local Buffer
000211
000212
000213  SIR_TBL         .WORD        SIR_NUM                   ;System Resource Table
000214  SIR_NUM         .BYTE        0                         ;SIR Number
000215                  .BYTE        0
000216                  .WORD        LP_INTR                   ;Address of interrupt handler
000217  INTR_BANK       .BYTE        0                         ;Bank number
000218  SIR_CNT         .EQU         *-SIR_NUM
000219                  .PAGE
000220  ;-----------------------------------------------------------------------
000221  ;
000222  ;       Hardware I/O Addresses
000223  ;
000224  ;-----------------------------------------------------------------------
000225
000226  B_REG           .EQU         0FFEF                     ;Bank Register
000227  E_REG           .EQU         0FFDF                     ;Environment Register
000228  ANYSLOT         .EQU         0FFDD                     ;Any Slot Interrupt Flag
000229
000230
000231  ;-----------------------------------------------------------------------
000232  ;
000233  ;       Universal Parallel Interface Card
000234  ;
000235  ;   Note:  to generate a STROBE (write to C082, or write to C080 with
000236  ;     AutoStrobe enabled), the CPU must be in Fixed Speed mode.  All
000237  ;     other registers may be accessed in either Fixed or Full Speed mode.
000238  ;
000239  ;-----------------------------------------------------------------------
000240
000241  PORTA           .EQU         0C080                     ;Output port A
000242  PORTB           .EQU         0C081                     ;Output port B
000243  READB           .EQU         0C083                     ;Input port
000244  STROBE          .EQU         0C082                     ;Send strobe
000245  DATARDY         .EQU         0C085                     ;Set DRDY, clear DRAK
000246  CLEAR           .EQU         0C087                     ;Clear DRDY, AutoStrobe
000247  ;  Set ACK, DRAK
000248
000249  STATUS          .EQU         0C084                     ;UPIC Status Register
000250
000251  ;    7        6        5        4        3        2        1        0
000252  ;--------+--------+--------+--------+--------+--------+--------+--------+
000253  ;   ACK  |  DRAK  |                                            |   ACK  |
000254  ; Latched| Latched|                                            |  Level |
000255  ;--------+--------+--------+--------+--------+--------+--------+--------+
000256
000257
000258  CTRLREG         .EQU         0C086                     ;UPIC Control Register
000259
000260  ;    7        6        5        4        3        2        1        0
000261  ;--------+--------+--------+--------+--------+--------+--------+--------+
000262  ;    Int Enable   |    Strobe/Ack Polarity   |    Strobe Width = 3 uS   |
000263  ;--------+--------+--------+--------+--------+--------+--------+--------+
000264  ;   ACK  |  DRAK  |   STB  |  DRDY  |ACK/DRAK|   +8   |   +4   |   -2   |
000265  ;--------+--------+--------+--------+--------+--------+--------+--------+
000266                  .PAGE
000267  ;-----------------------------------------------------------------------
000268  ;
000269  ;       Parallel Printer Driver -- Mainline
000270  ;

000271  ;-----------------------------------------------------------------------
000272
000273  LP_MAIN         SWITCH       REQCODE,8,LP_TBL
000274
000275  BADREQ          LDA          #XREQCODE                 ;Invalid request code
000276                  JSR          SYSERR
000277  NOTOPEN         LDA          #XNOTOPEN                 ;Device not open
000278                  JSR          SYSERR
000279
000280  LP_TBL          .WORD        LP_READ-1
000281                  .WORD        LP_WRITE-1
000282                  .WORD        LP_STAT-1
000283                  .WORD        LP_CNTL-1
000284                  .WORD        BADREQ-1
000285                  .WORD        BADREQ-1
000286                  .WORD        LP_OPEN-1
000287                  .WORD        LP_CLOSE-1
000288                  .WORD        LP_INIT-1
000289
000290
000291
000292  ;-----------------------------------------------------------------------
000293  ;
000294  ;       Parallel Printer Driver -- Initialization
000295  ;
000296  ;-----------------------------------------------------------------------
000297
000298  LP_INIT         .EQU         *
000299                  LDY          SLOT                      ;Check for valid slot #
000300                  DEY
000301                  CPY          #04                       ;If carry set out of range
000302                  RTS                                    ;SOS will mark inactive
000303                  .PAGE
000304  ;-----------------------------------------------------------------------
000305  ;
000306  ;       Parallel Printer Driver -- Open
000307  ;
000308  ;-----------------------------------------------------------------------
000309
000310  LP_OPEN         .EQU         *
000311                  BIT          OPENFLG                   ;Printer open?
000312                  BPL          $010                      ;No
000313                  LDA          #XNOTAVIL
000314                  JSR          SYSERR
000315
000316  $010            LDA          SLOT
000317                  ORA          #10
000318                  STA          SIR_NUM                   ;Set up resource number
000319                  ASL          A
000320                  ASL          A
000321                  ASL          A
000322                  ASL          A
000323                  STA          DEVOFF                    ;Offset to slot addresses
000324                  LDA          B_REG
000325                  AND          #0F
000326                  STA          INTR_BANK                 ;Set up interrupt bank
000327                  LDA          #SIR_CNT
000328                  LDX          SIR_TBL
000329                  LDY          SIR_TBL+1
000330                  JSR          ALLOCSIR                  ;Allocate resource
000331                  BCC          $020
000332                  LDA          #XNORESRC                 ;Can't allocate resource
000333                  JSR          SYSERR
000334
000335  $020            LDA          #IDLE
000336                  STA          CSTATE
000337                  JSR          CNTL00
000338                  LDA          #TRUE
000339                  STA          OPENFLG
000340                  RTS
000341                  .PAGE
000342  ;-----------------------------------------------------------------------
000343  ;
000344  ;       Parallel Printer Driver -- Close
000345  ;
000346  ;-----------------------------------------------------------------------
000347
000348  LP_CLOSE        .EQU         *
000349                  ASL          OPENFLG                   ;Printer open ?
000350                  BCS          $010                      ;Yes, closed now
000351                  JMP          NOTOPEN                   ;No, error request
000352
000353  $010            JSR          CNTL00                    ;Reset UPIC
000354                  LDA          #SIR_CNT
000355                  LDX          SIR_TBL
000356                  LDY          SIR_TBL+1
000357                  JSR          DEALCSIR                  ;Deallocate resource
000358                  RTS
000359
000360
000361
000362  ;-----------------------------------------------------------------------
000363  ;

000364  ;       Parallel Printer Driver -- Read
000365  ;
000366  ;-----------------------------------------------------------------------
000367
000368  LP_READ         .EQU         *
000369                  BIT          OPENFLG                   ;Printer open ?
000370                  BMI          $010                      ;Yes
000371                  JMP          NOTOPEN                   ;No, return error
000372
000373  $010            LDA          #XBADOP
000374                  JSR          SYSERR
000375                  .PAGE
000376  ;-----------------------------------------------------------------------
000377  ;
000378  ;       Parallel Printer Driver -- Write / Producer
000379  ;
000380  ;-----------------------------------------------------------------------
000381
000382  LP_WRITE        .EQU         *
000383                  BIT          OPENFLG                   ;Printer open ?
000384                  BMI          $010                      ;Yes
000385                  JMP          NOTOPEN                   ;No, return error
000386  $010            LDA          REQCNT                    ;One's Complement of REQCNT
000387                  EOR          #0FF                      ;  to implement main loop:
000388                  STA          REQCNT                    ;  While (RC:=RC+1)<>0 Do ...
000389                  LDA          REQCNT+1
000390                  EOR          #0FF
000391                  STA          REQCNT+1
000392
000393  $020            INC          REQCNT                    ;Increment byte count
000394                  BNE          $030
000395                  INC          REQCNT+1
000396                  BEQ          $090                      ;All done
000397  $030            LDY          #0
000398                  LDA          (BUFFER),Y                ;Get character
000399                  INC          BUFFER                    ;Increment buffer address
000400                  BNE          $040
000401                  INC          BUFFER+1
000402                  BNE          $040
000403                  LDX          #80
000404                  STX          BUFFER+1
000405                  INC          1401+BUFFER
000406  $040            BIT          PROPTNS                   ;Auto Line Feed?
000407                  BVC          $060                      ;  No
000408                  ASL          CRFLAG                    ;Last character CR?
000409                  BCC          $050                      ;  No
000410                  CMP          #ASC_LF                   ;This character LF?
000411                  BEQ          $020                      ;  Yes -- ignore it
000412  $050            CMP          #ASC_CR                   ;This character CR?
000413                  BNE          $060                      ;  No
000414                  ROR          CRFLAG                    ;  Yes -- set CR flag
000415  $060            LDX          PRODPTR
000416                  STA          LOCBUF,X                  ;Store character in local buffer
000417  $070            LDX          BUFCNT                    ;Check buffer count for max
000418                  INX
000419                  BEQ          $080
000420                  INC          PRODPTR                   ;Bump producer pointer
000421                  INC          BUFCNT                    ;  and buffer count
000422                  JMP          $020                      ;Loop
000423
000424  $080            JSR          PRIME                     ;Prime the consumer and
000425                  JMP          $070                      ;  recheck buffer count
000426
000427  $090            JSR          PRIME                     ;Make sure consumer is running
000428                  RTS
000429                  .PAGE
000430  ;-----------------------------------------------------------------------
000431  ;
000432  ;       Subroutine PRIME
000433  ;
000434  ;       Start up Consumer, and wait until printer is ready.
000435  ;
000436  ;-----------------------------------------------------------------------
000437
000438  PRIME           .EQU         *
000439                  BIT          CSTATE
000440                  BVC          $020
000441  $010            LDY          DEVOFF
000442                  LDA          READB,Y                   ;Wait for Ready
000443                  EOR          RDYSTAT
000444                  AND          RDYMASK
000445                  BNE          $010
000446  $020            PHP
000447                  SEI
000448                  JSR          LP_INTR                   ;Call interrupt handler
000449                  LDA          E_REG
000450                  AND          #BITOFF7
000451                  STA          E_REG                     ;Restore Full Speed
000452                  PLP
000453                  BIT          CSTATE
000454                  BVS          $010
000455                  RTS
000456

000457
000458
000459  ;-----------------------------------------------------------------------
000460  ;
000461  ;       Parallel Printer Driver -- Interrupt Handler / Consumer
000462  ;
000463  ;     This routine must be CALLED with INTERRUPTS DISABLED.  During
000464  ;     execution, it selects Fixed Speed mode and ENABLES INTERRUPTS.
000465  ;     It always EXITS in Fixed Speed mode with INTERRUPTS DISABLED.
000466  ;
000467  ;-----------------------------------------------------------------------
000468
000469  LP_INTR         LDA          E_REG
000470                  ORA          #BITON7                   ;Set Fixed Speed
000471                  STA          E_REG
000472                  LDY          DEVOFF                    ;Get offset to card addresses
000473                  LDA          CTRLWRD
000474                  AND          #BITOFF7                  ;Disable card's interrupt
000475                  STA          CTRLWRD
000476                  STA          CTRLREG,Y
000477                  LDA          #02
000478                  STA          ANYSLOT                   ;Clear 'Any Slot' interrupt
000479                  CLI
000480                  .PAGE
000481                  BIT          CSTATE                    ;Test Int.Handler State
000482                  BPL          $020                      ;  Send a character
000483                  LDX          STATUS,Y
000484                  BVC          $010                      ;  Waiting for ACK
000485                  LDA          READB,Y                   ;Check for READY
000486                  EOR          RDYSTAT
000487                  AND          RDYMASK
000488                  BEQ          $030                      ;Re-strobe printer
000489                  TXA
000490                  EOR          #BITON7
000491                  LDX          #WT_RDY                   ;Wait for Ready
000492                  BNE          $070                      ;  (unconditional branch)
000493
000494  $010            BPL          $050                      ;Wait for ACK
000495
000496  $020            LDA          BUFCNT                    ;Check buffer count
000497                  BEQ          $080                      ;Wait for data
000498                  LDX          CSMRPTR
000499                  LDA          LOCBUF,X                  ;Get character from buffer
000500                  EOR          INVERT
000501                  STA          PORTA,Y                   ;  and send it to printer
000502                  DEC          BUFCNT
000503                  INC          CSMRPTR
000504  $030            STA          STROBE,Y                  ;Stobe Printer
000505                  LDX          #WT_RDY
000506                  LDA          READB,Y                   ;Printer Ready?
000507                  EOR          RDYSTAT
000508                  AND          RDYMASK
000509                  BNE          $060                      ;  No
000510                  LDX          TIMEOUT
000511  $040            LDA          STATUS,Y                  ;Wait for ACK or Time Out
000512                  BMI          $020                      ;  ACK, send next character
000513                  DEX
000514                  BPL          $040
000515  $050            LDX          #WT_ACK                   ;Wait for ACK
000516
000517  $060            LDA          #BITON7                   ;Enable Interrupt
000518
000519  $070            SEI
000520                  STX          CSTATE                    ;Set Int.Handler State
000521                  AND          #BITON7                   ;Enable Interrupt with
000522                  ORA          CTRLWRD                   ;  bit 7 of Accumulator
000523                  STA          CTRLWRD
000524                  STA          CTRLREG,Y
000525                  RTS
000526
000527  $080            SEI
000528                  LDX          #IDLE
000529                  STX          CSTATE                    ;Wait for data
000530                  RTS
000531                  .PAGE
000532  ;-----------------------------------------------------------------------
000533  ;
000534  ;       Parallel Printer Driver -- Status
000535  ;
000536  ;-----------------------------------------------------------------------
000537
000538  LP_STAT         BIT          OPENFLG                   ;Printer open ?
000539                  BMI          $010
000540                  JMP          NOTOPEN                   ;No, return error
000541
000542  $010            LDY          #00
000543                  LDX          CTLSTAT
000544                  BEQ          STAT00
000545                  DEX
000546                  BEQ          STAT01
000547                  DEX
000548                  BEQ          STAT02
000549                  DEX

000550                  BEQ          STAT03
000551  BADCTL          LDA          #XCTLCODE                 ;Invalid control code
000552                  JSR          SYSERR
000553
000554  STAT00          RTS                                    ;0 -- NOP
000555
000556  STAT01          TYA                                    ;1 -- Status table
000557                  STA          (CSLIST),Y
000558                  RTS
000559
000560  STAT02          TYA                                    ;2 -- New line
000561                  STA          (CSLIST),Y
000562                  RTS
000563
000564  STAT03          LDX          DEVOFF                    ;3 -- Error status and buffer size
000565                  LDA          READB,X
000566                  EOR          RDYSTAT
000567                  STA          (CSLIST),Y
000568                  INY                                    ;Error status byte
000569                  LDA          #0FF                      ;Buffer size, low byte
000570                  STA          (CSLIST),Y
000571                  JSR          $010                      ; high byte
000572                  INY
000573                  LDA          BUFCNT                    ;Number of chars in buffer, low
000574                  STA          (CSLIST),Y
000575  $010            INY
000576                  LDA          #00                       ; high byte
000577                  STA          (CSLIST),Y
000578                  RTS
000579                  .PAGE
000580  ;-----------------------------------------------------------------------
000581  ;
000582  ;       Parallel Printer Driver -- Control
000583  ;
000584  ;-----------------------------------------------------------------------
000585
000586  LP_CNTL         BIT          OPENFLG                   ;Printer open ?
000587                  BMI          $010                      ;Yes
000588                  JMP          NOTOPEN
000589
000590  $010            LDX          CTLSTAT
000591                  BEQ          CNTL00
000592                  DEX
000593                  BEQ          CNTL01
000594                  DEX
000595                  BEQ          CNTL02
000596                  JMP          BADCTL                    ;Invalid request number
000597
000598  CNTL00          .EQU         *
000599                  BIT          CSTATE                    ;Test consumer state
000600                  BPL          $020                      ;  Ready
000601  $010            JSR          PRIME                     ;Prime consumer and
000602                  BIT          CSTATE                    ;  recheck state
000603                  BMI          $010
000604  $020            LDA          #00
000605                  STA          BUFCNT                    ;Set buffer count to 0
000606                  STA          PRODPTR                   ;Set producer pointer
000607                  STA          CSMRPTR                   ;Set consumer pointer
000608                  STA          CRFLAG                    ;Set CR switch false
000609                  BIT          PROPTNS                   ;Check invert bit and
000610                  BPL          $030
000611                  LDA          #0FF
000612  $030            STA          INVERT                    ;  set INVERT value
000613                  PHP
000614                  SEI
000615                  LDY          DEVOFF
000616                  LDA          CTRLVAL
000617                  AND          #2F
000618                  STA          CTRLWRD                   ;Init UPIC control register
000619                  STA          CTRLREG,Y
000620                  STA          CLEAR,Y                   ;Disable auto strobe
000621                  PLP
000622                  RTS
000623
000624
000625  CNTL01          RTS
000626
000627
000628  CNTL02          RTS                                    ;2 New line
000629                  .END
000630

; ####################################################################################################
; #   END OF FILE:  PARPRINT.text
; #   LINES      :  630
; #   CHARACTERS :  32172
; #   Formatter  :  Assembly Language Reformatter 1.0.2 (07 January 1998)
; #   Author     :  David T. Craig -- 71533.606@compuserve.com -- Santa Fe, New Mexico USA
; ####################################################################################################

###

